

Rekordok

Tartalomjegyzék

Rekord típus.....	2
Kérdések, feladatok.....	2
Példaprogram.....	3
Kérdések, feladatok.....	4

Készítette: Gál Tamás

Közzétéve: <http://informatika.gtportal.eu/>

[Creative Commons](#) -Nevezd meg!-Ne add el!-Így add tovább! 2.5 Magyarország licenc alatt használható

Rekord típus

A rekord strukturált adattípus, amelyben különböző típusú adatokat (mezőket) fogunk össze.

Segítségével együtt kezelhetjük az összetartozó adatokat. Pl. egy személy adatai, egy termék adatai...

Pascal program esetén a globális változók deklarációja előtt deklarálhatunk saját adattípusokat, így rekordokat is.

Rekord deklaráció:

```
type RekordNev = record
    MezoNev1: típus;
    MezoNev2: típus;
    ...
    MezoNevN: típus;
end;
```

Példa:

```
type DolgozoTip = record
    Nev: string[20];
    Hazas: boolean;
    Eletkor: byte;
end;
var Dolgozok: array[1..5]of DolgozoTip;
    d1, d2: DolgozoTip;
```

Hivatkozás a rekord egy mezőjére:

RekordNev.MezoNev1

A rekordváltozó neve után ponttal elválasztva írjuk a mező azonosítóját

d1.Nev
Dolgozok[5].Hazas

Ha rekordokat tömbben tárolunk, akkor nagyon hamar kevésnek bizonyulhat a rendelkezésre memória. Ekkor különösen fontos, hogy egy adat tárolására ne használjunk a szükségesnél több memóriát. Pl. élő személyek korának tárolására bőben elég Byte-ot használni. A neveknél sincs szükségünk 255 karakter hosszú string típusú mezőkre. Az adott feladat ismeretében azonban ki kell zárni azt az esetet, amikor a tárolni kívánt adat nem fér el a rendelkezésre álló helyen. Esetünkben a 20 karakter hosszú string lehet, hogy kevés lesz.

Kérdések, feladatok

- Deklarálj egy rekord típusú változót, amelyben könyveid címét, szerzőjét, árát és kiadásának évét tárolod!
- A rekord egyszerű vagy összetett adattípus?

Példaprogram

```
program RekordPl;
type DolgozoTip = record
    Nev: string[20];
    Eletkor: byte;
end;
const n=3;
var Dolgozok: array[1..n]of DolgozoTip;
    d1, d2: DolgozoTip;

procedure Feltolt;
var i:integer;
begin
    for i:=1 to n do begin
        write('Az ',i,'. dolgozó neve:');
        readln(Dolgozok[i].Nev );
        write('Az ',i,'. dolgozó kora:');
        readln(Dolgozok[i].Eletkor );
    end
end;

procedure Kiir;
var i:integer;
begin
    writeln();
    writeln('Dolgozóink');
    for i:=1 to n do begin
        writeln('Az ',i,'. dolgozó neve:',Dolgozok[i].Nev);
        writeln('Az ',i,'. dolgozó kora:',Dolgozok[i].Eletkor );
    end;
end;

procedure Csere(var a, b: DolgozoTip); // cím szerinti paraméterátadás
var tmp:DolgozoTip;
begin
    tmp := a;
    a := b;
    b := tmp;
end;

procedure Rendez;
var i, j:integer;
begin
    j:=1;
    for i:=n-1 downto 1 do begin
        for j:=1 to i do begin
            if Dolgozok[j].Nev > Dolgozok[j+1].Nev then Csere(Dolgozok[j],Dolgozok[j+1]);
        end;
    end;
end;

begin
    Feltolt;
    Kiir;
    Rendez;
    Kiir;
end.
```

Kérdések, feladatok

- Készítsd el a példaprogram folyamatábráját és mondatszerű leírását!
- Melyik programozási tételt használjuk a 'Rendez' nevű eljárásnál?
- Mit hasonlítunk össze és mit cserélünk ki a 'Rendez' nevű eljárásban?
- Készítsd el annak a programnak a mondatszerű leírását, amely képes 10 db autó típusát, rendszámát és évjárátát megjeleníteni. Az adatok megadását tedd lehetővé a felhasználó számára.